Lab 3

23&24. We refer to Section 3 of the tutorial. These functions are defined piecewise.

```
> f23:=piecewise(t<=2, t, 3-t);
```

$$f23 := \begin{cases} t & t \le 2 \\ 3 - t & \mathit{otherwise} \end{cases} \tag{1}$$

```
> plot(f23,t=-2..4,discont=true);
```



We deduce that this function is not continuous in t=2. Note that, in order to solve this exercise, it is not necessary to define f23 as a function. It is enough if we work with its expression.

```
> f24:=piecewise(t>0 and t<Pi, t, t>Pi, t*exp(Pi-t));
```

$$f24 := \begin{cases} t & 0 < t \textbf{ and } t < \pi \\ t\,e^{\pi - t} & \pi < t \end{cases} \tag{2}$$

```
> plot(f24,t=-2..10,discont=true);
```



We deduce that this function is continuous, but it seems that t=Pi is an angular point, that is, the function is not differentiable in t=Pi.

25.

```
> sol:=dsolve({diff(x(t),t$2)+x(t)=f24,x(0)=0,D(x)(0)=1}, x(t));
```

$$sol := x(t) = \begin{cases} \sin(t) & t < 0 \\ t & t < \pi \\ -\sin(t) - \frac{1}{2}\sin(t)\,\pi - \frac{1}{2}\cos(t)\,\pi + \frac{1}{2}\,t\,e^{\pi - t} + \frac{1}{2}\cos(t) + \frac{1}{2}\,e^{\pi - t} & \pi \le t \end{cases} \quad \textbf{(3)}$$

```
> plot(rhs(sol),t=-2..10);
```



Note that the DE that we solved is linear nonhomogeneous with constant coefficients and that the nonhomogeneous term, f24, is a continuous function, as in the theoretical framework established at the lecture. Thus a solution is a C^2 function. We can also see that its graph is "smooth".

26.

```
> sol:=dsolve({diff(x(t),t$2)+x(t)=cos(omega*t),x(0)=0,D(x)(0)=0},x(t));
```

$$sol := x(t) = \frac{\cos(t)}{\omega^2 - 1} - \frac{\cos(\omega t)}{\omega^2 - 1} \tag{4}$$

```
> phi:=unapply(rhs(sol),t,omega);
```

$$\phi := (t, \omega) \rightarrow \frac{\cos(t)}{\omega^2 - 1} - \frac{\cos(\omega t)}{\omega^2 - 1} \tag{5}$$

```
> phi(t,2);
```

$$\frac{1}{3}\cos(t) - \frac{1}{3}\cos(2\,t) \tag{6}$$

```
> phi(t,1);
Error, (in phi) numeric exception: division by zero
> limit(phi(t,omega),omega=1);
```

$$\frac{1}{2}\, t \sin(t) \tag{7}$$

```
> sol1:=dsolve({diff(x(t),t$2)+x(t)=cos(t),x(0)=0,D(x)(0)=0},x(t));
```

$$sol1 := x(t) = \frac{1}{2}\, t \sin(t) \tag{8}$$

```
> phi1:=rhs(sol1);
```

$$\phi 1 := \frac{1}{2}\, t \sin(t) \tag{9}$$

```
> plot(phi1,t=-10..10); plot(phi1,t=-infinity..infinity);
```





It seems that phi(t,1) is oscillating with amplitude that goes to infinity as t goes to infinity. Justify rigourously in your notebooks that this is indeed the situation.
Note that this function is a solution of the DE x''+x=cos(t), thus the eternal force is bounded. So, you put a bounded force, but you receive an unbounded response. This is because of the **Resonance**: the frequency of the external force (here cos(t)) is equal to the internal frequency of the unforced oscillator (x''+x=0, x=c1 cos(t)+ c2 sin(t)).

27.

```
> sol27:=dsolve({diff(x(t),t$2)-4*x(t)=exp(alpha*t),x(0)=0,D(x)(0)=
  0},x(t));
```

$$sol27 := x(t) = -\frac{1}{4}\,\frac{e^{2t}}{\alpha-2} + \frac{1}{4}\,\frac{e^{-2t}}{\alpha+2} + \frac{e^{\alpha t}}{\alpha^2-4} \quad \textbf{(10)}$$

```
> phi27:=unapply(rhs(sol27),t,alpha);
```

$$\phi 27 := (t,\alpha) \rightarrow -\frac{1}{4}\,\frac{e^{2t}}{\alpha-2} + \frac{1}{4}\,\frac{e^{-2t}}{\alpha+2} + \frac{e^{\alpha t}}{\alpha^2-4} \quad \textbf{(11)}$$

```
> simplify(limit(phi27(t,alpha),alpha=2));
```

(12)

$$\frac{1}{4}\, t\, e^{2t} - \frac{1}{16}\, e^{2t} + \frac{1}{16}\, e^{-2t} \tag{12}$$

```
> dsolve({diff(x(t),t$2)-4*x(t)=exp(2*t),x(0)=0,D(x)(0)=0},x(t));
```

$$x(t) = \frac{1}{4}\, t\, e^{2t} - \frac{1}{16}\, e^{2t} + \frac{1}{16}\, e^{-2t} \tag{13}$$

We note that the solution is continuous in alpha=2, too.

28.

```
> dsolve({diff(x(t),t)-x(t)=0,x(0)=1},x(t));
```

$$x(t) = e^{t} \tag{14}$$

```
> dsolve({diff(x(t),t)-x(t)=0,x(0)=1},x(t),series); series(exp(t),
  t);
```

$$x(t) = 1 + t + \frac{1}{2}\, t^2 + \frac{1}{6}\, t^3 + \frac{1}{24}\, t^4 + \frac{1}{120}\, t^5 + \mathrm{O}(t^6)$$

$$1 + t + \frac{1}{2}\, t^2 + \frac{1}{6}\, t^3 + \frac{1}{24}\, t^4 + \frac{1}{120}\, t^5 + \mathrm{O}(t^6) \tag{15}$$

```
> Order:=15;
```

$$Order := 15 \tag{16}$$

```
> dsolve({diff(x(t),t)-x(t)=0,x(0)=1},x(t),series); series(exp(t),
  t);
```

$$x(t) = 1 + t + \frac{1}{2}\, t^2 + \frac{1}{6}\, t^3 + \frac{1}{24}\, t^4 + \frac{1}{120}\, t^5 + \frac{1}{720}\, t^6 + \frac{1}{5040}\, t^7 + \frac{1}{40320}\, t^8 + \frac{1}{362880}\, t^9 + \frac{1}{3628800}\, t^{10} + \frac{1}{39916800}\, t^{11} + \frac{1}{479001600}\, t^{12} + \frac{1}{6227020800}\, t^{13} + \frac{1}{87178291200}\, t^{14} + \mathrm{O}(t^{15})$$

$$1 + t + \frac{1}{2}\, t^2 + \frac{1}{6}\, t^3 + \frac{1}{24}\, t^4 + \frac{1}{120}\, t^5 + \frac{1}{720}\, t^6 + \frac{1}{5040}\, t^7 + \frac{1}{40320}\, t^8 + \frac{1}{362880}\, t^9 + \frac{1}{3628800}\, t^{10} + \frac{1}{39916800}\, t^{11} + \frac{1}{479001600}\, t^{12} + \frac{1}{6227020800}\, t^{13} + \frac{1}{87178291200}\, t^{14} + \mathrm{O}(t^{15}) \tag{17}$$

This was just a toy-example.

29.

```
> dsolve({x^2*diff(u(x),x$2)+x*diff(u(x),x)+x^2*u(x)=0,u(0)=1,D(u)
  (0)=0},u(x));
```

$$u(x) = \mathrm{BesselJ}(0, x) \tag{18}$$

```
> dsolve({x^2*diff(u(x),x$2)+x*diff(u(x),x)+x^2*u(x)=0,u(0)=1,D(u)
  (0)=0},u(x),series);
```

$$u(x) = 1 - \frac{1}{4}\, x^2 + \frac{1}{64}\, x^4 - \frac{1}{2304}\, x^6 + \frac{1}{147456}\, x^8 - \frac{1}{14745600}\, x^{10} + \frac{1}{2123366400}\, x^{12} \tag{19}$$

$$-\frac{1}{416179814400}\,x^{14}+O(x^{15})$$

The solution is a special function, which can not be written as a finite combination of elementary functions. Thus, we do not have an expression of it to compute the coefficients of its Taylor series around x=0. But the coefficients can be computed using the differential equation. Note that u(0)=1 and u'(0)=0 are given, but one can obtain u''(0) using the DE. More exactly, after passing to the limit as x->0 in u''(x)+u'(x)/x+u(x)=0 we obtain 2u''(0)+u(0)=0. Thus, u''(0)= - 1/2.
Taking the derivative with respect to x in the DE u''(x)+u'(x)/x+u(x)=0 we get u'''(x)+u''(x)/x-u'(x)/x^2+u'(x)=0. Taking x->0 we have u'''(0)+u'''(0)/2+u'(0)=0. Thus, u'''(0)=0. Continuing this procedure, one can obtain inductively all the coefficients of the Taylor expansion around x=0 of the solution.

There is also another procedure, maybe simpler: to write the solution as a power series with unknown coefficients u(x)=a0+a1 x+a2 x^2+a3 x^3 + ... , then substitute it in the DE, collect the similar terms to obtain one power series =0. Finally equate with 0 each coefficient of this last series. Using that a0=1 and a1=0, one can obtain inductively the other coefficients.

30.

```
> dsolve(diff(y(x),x$2)-2*x*diff(y(x),x)+4*y(x)=0,y(x),series);
```

$$y(x)=y(0)+D(y)(0)\,x-2\,y(0)\,x^2-\frac{1}{3}\,D(y)(0)\,x^3-\frac{1}{30}\,D(y)(0)\,x^5-\frac{1}{210}\,D(y)(0)\,x^7-\frac{1}{1512}\,D(y)(0)\,x^9-\frac{1}{11880}\,D(y)(0)\,x^{11}-\frac{1}{102960}\,D(y)(0)\,x^{13}+O(x^{15}) \quad \textbf{(20)}$$

Indeed, taking y'(0)=0 it seems that all the terms, but 2 of them, cancel. Thus, it seems that y(x)= C (1 -2x^2) is a solution for arbitrary real C.

```
> ysol:=C*(1-2*x^2); diff(ysol,x$2)-2*x*diff(ysol,x)+4*ysol; expand
  (%);
```

$$ysol := C\,(-2\,x^2+1)$$

$$-4\,C+8\,x^2\,C+4\,C\,(-2\,x^2+1)$$

$$0 \quad \textbf{(21)}$$

It is a solution, indeed.

32.

```
> sol32:=dsolve({diff(x(t),t)=-2*x(t),diff(y(t),t)=-3*y(t),x(0)=1,y
  (0)=1},{x(t),y(t)});
```

$$sol32 := \{x(t)=e^{-2\,t},\,y(t)=e^{-3\,t}\} \quad \textbf{(22)}$$

```
> phi32x:=rhs(sol32[1]); phi32y:=rhs(sol32[2]);
```

$$phi32x := e^{-2\,t}$$

$$phi32y := e^{-3\,t} \quad \textbf{(23)}$$

```
> plot([phi32x,phi32y,t=0..2]);
```



Note that, at the initial time t=0, we are in the point (1,1) and then we move very fast toward (0,0).

33.

```
> sol33:=dsolve({diff(x(t),t)=-2*x(t),diff(y(t),t)=3*y(t),x(0)=1,y
  (0)=1},{x(t),y(t)});
```

$$sol33 := \{x(t) = e^{-2t}, y(t) = e^{3t}\} \quad \textbf{(24)}$$

```
> phi33x:=rhs(sol33[1]); phi33y:=rhs(sol33[2]);
```

$$phi33x := e^{-2t}$$
$$phi33y := e^{3t} \quad \textbf{(25)}$$

```
> plot([phi33x,phi33y,t=0..2]);
```



Note that this looks like a saddle. At the initial time t=0, we are in the point (1,1) and then we move left and then very fast up.

34.

```
> sol34:=dsolve({diff(x(t),t)=-y(t),diff(y(t),t)=4*x(t),x(0)=1,y(0)=1},{x(t),y(t)});
```

$$sol34 := \left\{x(t) = -\frac{1}{2}\sin(2\,t) + \cos(2\,t),\, y(t) = \cos(2\,t) + 2\sin(2\,t)\right\} \quad \textbf{(26)}$$

```
> phi34x:=rhs(sol34[1]); phi34y:=rhs(sol34[2]);
```

$$phi34x := -\frac{1}{2}\sin(2\,t) + \cos(2\,t)$$

$$phi34y := \cos(2\,t) + 2\sin(2\,t) \quad \textbf{(27)}$$

```
> plot([phi34x,phi34y,t=0..2]); plot([phi34x,phi34y,t=0..Pi]);
```


2
1
−1
−0.5
0
0.5
1
−1
−2



It seems that this is an ellipse. At the initial time t=0, we are in the point (1,1) and then we encircle the origin in the trigonometric sense to come back after T=Pi at the initial position. After, we repeat and move again on the same trajectory (orbit).

35.

```
> sol35:=dsolve({diff(x(t),t)=-x(t)+3*y(t),diff(y(t),t)=-3*x(t)-y
  (t),x(0)=1,y(0)=1},{x(t),y(t)});
```

$$sol35 := \{x(t) = e^{-t}(\cos(3t) + \sin(3t)), y(t) = e^{-t}(\cos(3t) - \sin(3t))\} \quad \textbf{(28)}$$

```
> phi35x:=rhs(sol35[1]); phi35y:=rhs(sol35[2]);
```

$$phi35x := e^{-t}(\cos(3t) + \sin(3t))$$

$$phi35y := e^{-t}(\cos(3t) - \sin(3t)) \quad \textbf{(29)}$$

```
> plot([phi35x,phi35y,t=0..10]);
```



Note that, at the initial time t=0, we are in the point (1,1) and then we move toward (0,0) encircling the origin.

36.

```
> sol36:=dsolve({diff(x(t),t)=-x(t)+3*y(t),diff(y(t),t)=3*x(t)-y
  (t),x(0)=1,y(0)=1},{x(t),y(t)});
```

$$sol36 := \{x(t) = e^{2t}, y(t) = e^{2t}\} \quad \textbf{(30)}$$

```
> phi36x:=rhs(sol36[1]); phi36y:=rhs(sol36[2]);
```

$$phi36x := e^{2t}$$

$$phi36y := e^{2t} \quad \textbf{(31)}$$

```
> plot([phi36x,phi36y,t=0..10]); plot([phi36x,phi36y,t=0..0.1]);
```



4. × 10^8
3. × 10^8
2. × 10^8
1. × 10^8
0
0
1. × 10^8
2. × 10^8
3. × 10^8
4. × 10^8



>

Note that, at the initial time t=0, we are in the point (1,1) and then we move very fast toward infinity. Note also that, there is no big difference between the differential systems 35 and 36, while the initial conditions are the same. Nevertheless, the trajectories are very very different.

Can you imagine that it is possible to know the shape of the orbits of a linear system (like the ones above) computing only the eigenvalues of the matrix of the system? So, without knowing the exact solutions!