```
>
```

Lab 4

1.

```
> eq1:=diff(x(t),t)=1-x(t)^2;
```

$$eq1 := \frac{d}{dt} x(t) = 1 - x(t)^2 \quad (1)$$

```
> solve(1-x^2=0,x);
```

$$-1, 1 \quad (2)$$

These are the two constant solutions of the given differential equation x'=1-x^2.

```
> solexpr1:=rhs(dsolve({eq1,x(0)=eta}));
```

$$solexpr1 := \tanh(t + \operatorname{arctanh}(\eta)) \quad (3)$$

This is the expression of the solution.

```
> phi:=unapply(solexpr1,(t,eta));
```

$$\phi := (t, \eta) \to \tanh(t + \operatorname{arctanh}(\eta)) \quad (4)$$

the command "unapply" creates a function of the given expression solexpr1, having the mentioned variables (t,eta).

```
> phi(t,eta);
```

$$\tanh(t + \operatorname{arctanh}(\eta)) \quad (5)$$

This is just to check that, indeed, the expression of the function is what we wanted.

```
> phi(t,1);
Error, (in arctanh) numeric exception: division by zero
> phi(t,-1);
Error, (in arctanh) numeric exception: division by zero
```

Indeed, the expression of phi is not well defined for eta=1 and eta= -1. It is easy to notice that phi(t,1)= 1 and phi(t,-1)=-1 for any real t. Anyway, we find them using Maple.

```
> dsolve({eq1,x(0)=1}); dsolve({eq1,x(0)=-1});
```

$$x(t) = 1$$

$$x(t) = -1 \quad (6)$$

Now we want to plot the graphs.

First we take the initial value, eta, in the interval (-1,1).

```
> plot({phi(t,-0.7),phi(t,-0.5),phi(t,0.4),phi(t,0.9)},t=-5..5);
```




```
> plot({phi(t,-0.7),phi(t,-0.5),phi(t,0.4),phi(t,0.9)},t=-infinity.
  .infinity);
```



It seems that all these functions have the same limits as t->infinity and, respectively, as t-> - infinity. We compute these limits using Maple.

```
> assume(-1<eta and eta<1);  limit(phi(t,eta),t=infinity); limit
  (phi(t,eta),t=-infinity);
```

$$1$$

$$-1 \tag{7}$$

Write in your notebook what we just have found.

Do not forget that we have assumptions on eta. Anyway, we are allowed to forget, since we can check with Maple.

```
> hasassumptions(eta);
```

$$\mathit{true} \tag{8}$$

```
> eta:='eta'; hasassumptions(eta);
```

$$\eta := \eta$$

$$\mathit{false} \tag{9}$$

Now we plot some functions phi(t,eta) with eta, in the interval (1,infinity). But still eta very close to 1.
We also choose small intervals for t.
If you want to see what happen if you consider, also, bigger values of eta, or big intervals for t, just try.

```
> plot({phi(t,1.1),phi(t,1.2),phi(t,1.3)},t=0..2);
```




```
> plot(phi(t,1.1),t=-1.5..5);
```



It seems that these functions have the same limit as t->infinity. We compute this with Maple. In addition, it seems that phi(t,1.1) explodes in a point near t= - 1.2

```
> assume(1<eta);  limit(phi(t,eta),t=infinity); eta='eta';
```

$$1$$

$$\eta\sim = \eta \quad (10)$$

Write in your notebook what we just have found.

Now we plot some functions phi(t,eta) with eta, in the interval (-infinity, -1). But still eta very close to -1. We also choose small intervals for t.

```
> plot({phi(t,-1.1),phi(t,-1.2),phi(t,-1.3)},t=-2..0);
```



It seems that these functions have the same limit as t-> - infinity. We compute this with Maple.

```
> assume(eta<-1);  limit(phi(t,eta),t=-infinity); eta='eta';
```

$$-1$$

$$\eta\sim = \eta \tag{11}$$

Write in your notebook what we just have found.

In addition, from the graphs that we have, it seems that, indeed, any nonconstant solution is strictly monotone.

Finally, in your notebooks represent the phase portrait of x'=1-x^2 and confirm the properties you found.

2.

```
> eq2:=diff(x(t),t)=1-t*x(t)^3;
```

$$eq2 := \frac{\mathrm{d}}{\mathrm{d}t}\, x(t) = 1 - t\, x(t)^3 \tag{12}$$

```
> dsolve(eq2); dsolve({eq2, x(0)=0},x(t));
```

So, Maple does not return anything. We will increase the level of informations for "dsolve" to see if

Maple just ignored us :-)

```
> infolevel[dsolve]:=3;
```

$$infolevel_{dsolve} := 3 \tag{13}$$

```
> dsolve(eq2);
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
trying Bernoulli
trying separable
trying inverse linear
trying homogeneous types:
trying Chini
differential order: 1; looking for linear symmetries
trying exact
trying Abel
trying inverse_Riccati
differential order: 1; trying a linearization to 2nd order
--- trying a change of variables {x -> y(x), y(x) -> x}
differential order: 1; trying a linearization to 2nd order
trying 1st order ODE linearizable_by_differentiation
--- Trying Lie symmetry methods, 1st order ---
 -> Computing symmetries using: way = 3
 -> Computing symmetries using: way = 4
 -> Computing symmetries using: way = 2
trying symmetry patterns for 1st order ODEs
-> trying a symmetry pattern of the form [F(x)*G(y), 0]
-> trying a symmetry pattern of the form [0, F(x)*G(y)]
-> trying symmetry patterns of the forms [F(x),G(y)] and [G(y),F
(x)]
-> trying a symmetry pattern of the form [F(x),G(x)]
-> trying a symmetry pattern of the form [F(y),G(y)]
-> trying a symmetry pattern of the form [F(x)+G(y), 0]
-> trying a symmetry pattern of the form [0, F(x)+G(y)]
-> trying a symmetry pattern of the form [F(x),G(x)*y+H(x)]
-> trying a symmetry pattern of conformal type
```

It seems that Maple tried really hard to solve this equation!!!

So, it was proved that there exists a unique solution of the IVP x'=1-t*x^3, x(0)=0 but we can not see it! We have just limited posibilities to write down an expression.

```
> Order:=25; dsolve({eq2, x(0)=0},x(t),series);
```

$$Order := 25$$

$$x(t) = t - \frac{1}{5}t^5 + \frac{1}{15}t^9 - \frac{8}{325}t^{13} + \frac{263}{27625}t^{17} - \frac{6583}{1740375}t^{21} + \mathrm{O}(t^{25}) \tag{14}$$

Anyway, as we explained in the previous Lab, Maple can find, step by step, the Taylor polynomials that approximate the solution of the IVP x'=1-t*x^3, x(0)=0.

```
> with(DEtools):
```

```
> dfieldplot(eq2,x(t),t=-4..4,x=-1..1);
```



From the direction field you can "guess" the shape of the solution curves of eq2.

```
> DEplot(eq2,x(t),t=-4..4,[x(0)=0]);
```



The graph of the solution of eq2 that satifies x(0)=0 looks like in the above picture. It follows the directions indicated by the arrows. Roughly speaking, this is how it was drawn. But behind there are computations done using a numerical formula.

In the following we fix 3 solutions by fixing their initial values.

```
> DEplot(eq2,x(t),t=-4..4,[x(0)=0,x(0)=-0.5,x(0)=0.5]);
```



3.

```
> eq3a:=diff(y(x),x)=y(x)/x;
```

$$eq3a := \frac{\mathrm{d}}{\mathrm{d}x}\, y(x) = \frac{y(x)}{x} \tag{15}$$

```
> dfieldplot(eq3a, y(x), x=-5..5, y=-5..5);
```



It seems that the integral curves are straight lines passing through the origin. This is confirmed after we see the general solution.

```
> dsolve(eq3a);
```

$$y(x) = \_C1\, x \tag{16}$$

```
> eq3b:=diff(y(x),x)=-y(x)/x;
```

$$eq3b := \frac{d}{dx}\, y(x) = -\frac{y(x)}{x} \tag{17}$$

```
> dfieldplot(eq3b, y(x), x=-5..5, y=-5..5);
```



It seems that the integral curves are hyperbolas whose asymptotes are the axes of coordinates. This is confirmed after we see the general solution.

```
> dsolve(eq3b);
```

$$y(x) = \frac{\_C1}{x} \tag{18}$$

```
> eq3c:=diff(y(x),x)=-x/y(x);dfieldplot(eq3c, y(x), x=-5..5, y=-5..5);
```

$$eq3c := \frac{d}{dx}\, y(x) = -\frac{x}{y(x)}$$



It seems that the integral curves are circles centered in the origin. This is confirmed after we see the general solution.

```
> dsolve(eq3c);
```

$$y(x) = \sqrt{-x^2 + \_C1}, y(x) = -\sqrt{-x^2 + \_C1} \quad \textbf{(19)}$$

```
> eq3d:=diff(y(x),x)=-2*x/y(x);dfieldplot(eq3d, y(x), x=-5..5, y=
  -5..5);
```

$$eq3d := \frac{d}{dx}\, y(x) = -\frac{2\,x}{y(x)}$$



It seems that the integral curves are ellipses. This is confirmed after we see the general solution.

```
> dsolve(eq3d);
```

$$y(x) = \sqrt{-2\,x^2 + \_C1},\, y(x) = -\sqrt{-2\,x^2 + \_C1} \tag{20}$$

4.

```
> with(plots):
> contourplot(x*y,x=-5..5,y=-5..5);
```



The level curves of H1(x,y)=x*y are hyperbolas whose asymptotes are the axes of coordinates. Note that they look like the solution curves of eq3b y'=-y/x whose general solution is x*y=c, c arbitrary real constant.

```
> diff(x*y(x),x)=0;
```

$$y(x) + x\left(\frac{d}{dx}\, y(x)\right) = 0 \tag{21}$$

This is exactly eq3b.

```
> contourplot(x^2+y^2,x=-5..5,y=-5..5);
```



The level curves of H2(x,y)=x^2+y^2 are circles centered in the origin. Note that they look like the solution curves of eq3c y'=-x/y whose general solution is x^2+y^2=c, c arbitrary real constant.

```
> diff(x^2+y(x)^2,x)=0;
```

$$2\,x+2\left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right)y(x)=0 \tag{22}$$

This is exactly eq3c.

```
> contourplot(2*x^2+y^2,x=-5..5,y=-5..5);
```



The level curves of H3(x,y)=2*x^2+y^2 are ellipses around the origin. Note that they look like the solution curves of eq3d y'=-2x/y whose general solution is 2x^2+y^2=c, c arbitrary real constant.

```
> diff(2*x^2+y(x)^2,x)=0;
```

$$4\,x+2\left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right)y(x)=0 \tag{23}$$

This is exactly eq3d.

5.

```
> diff(x^2+4*y(x)^2,x)=0;
```

$$2\,x+8\left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right)y(x)=0 \tag{24}$$

The DE is y'=-x/(4y).

6.

```
> contourplot(y^2-cos(x),x=-0.5..0.5,y=-0.5..0.5);
```



Indeed, in this small region around the origin, the level curves seems to be closed. It can be proved analytically that this is the case.

>