Lab 5

Problem **I.1.** Study the system x'=2x-x^2-xy, y'=-y+xy .

a)

```
> solve({2*x-x^2-x*y=0,-y+x*y=0});
```
Note that this nonlinear system has three equilibria (0,0), (2,0) and (1.1).

$$\{x=0, y=0\}, \{x=2, y=0\}, \{x=1, y=1\} \quad \textbf{(1)}$$

```
> with(linalg):with(DEtools):with(VectorCalculus):
```
b)

```
> Jm1:=Jacobian([ 2*x-x^2 - x*y, -y + x*y ],[x,y]);
```

$$Jm1 := \begin{bmatrix} -2x-y+2 & -x \\ y & x-1 \end{bmatrix} \quad \textbf{(2)}$$

```
> A1:=subs([x=0,y=0],Jm1);
```
So, the linearization of the given nonlinear system around the equilibrium point (0,0) is X'=A1 X, that is x'=2x, y'= -y.

$$A1 := \begin{bmatrix} 2 & 0 \\ 0 & -1 \end{bmatrix} \quad \textbf{(3)}$$

```
> eigenvalues(A1);
```
Note that both eigenvalues are real, different from zero. Then the equilibrium point (0,0) is hyperbolic and we can apply the Linearization method (Theorem 3 from Sem5). In this case the linear system X'=A1 X has a saddle, thus the equil (0,0) of the nonlinear system is unstable.

$$2, -1 \quad \textbf{(4)}$$

```
> A2:=subs([x=2,y=0],Jm1);
```
So, the linearization of the given nonlinear system around the equilibrium point (2,0) is X'=A2 X, that is x'= -2x-2y, y'=y.

$$A2 := \begin{bmatrix} -2 & -2 \\ 0 & 1 \end{bmatrix} \quad \textbf{(5)}$$

```
> eigenvalues(A2);
```
Note that both eigenvalues are real, different from zero. Then the equilibrium point (2,0) is hyperbolic and we can apply the Linearization method (Theorem 3 from Sem5). In this case the linear system X'=A2 X has a saddle, thus the equil (2,0) of the nonlinear system is unstable.

$$-2, 1 \quad \textbf{(6)}$$

```
> A3:=subs([x=1,y=1],Jm1);
```
So, the linearization of the given nonlinear system around the equilibrium point (1,1) is X'=A3 X, that is x'= -x-y, y'=x.

$$A3 := \begin{bmatrix} -1 & -1 \\ 1 & 0 \end{bmatrix} \quad \textbf{(7)}$$

```
> eigenvalues(A3);
```
Note that both eigenvalues are complex conjugate, whose real part is different from zero. Then the equilibrium point (1,1) is hyperbolic and we can apply the Linearization method (Theorem 3 from Sem5). Since the real part of the eigenvalues is stricly negative, the linear system X'=A3 X has an attracting focus, thus the equilibrium (1,1) of the nonlinear system is also an attractor.

$$-\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\sqrt{3}, -\frac{1}{2}-\frac{1}{2}\,\mathrm{I}\sqrt{3} \quad \textbf{(8)}$$

```
> dfieldplot([diff(x(t),t)=2*x(t)-x(t)^2 - x(t)*y(t),diff(y(t),t)=
-y(t) + x(t)*y(t)],[x(t),y(t)], t=0..1, x=-3..3,y=-3..3);
```
This is the direction field in the box [-3,3]x[-3,3]. Indeed, we can notice that, around the equilibrium points (0,0), (2,0) and (1,1) the direction field is not so regular.




```
> dfieldplot([diff(x(t),t)=2*x(t)-x(t)^2 - x(t)*y(t),diff(y(t),t)=
-y(t) + x(t)*y(t)],[x(t),y(t)], t=0..1, x=-0.5..0.5,y=-0.5..0.5);
```
This is the direction field in the box [-0.5,0.5]x[-0.5,0.5]. The only equilibrium point in this box is (0,0). It seems that the orbits in this small box looks like the orbits of a linear system with a saddle. In the next figure we will see the direction field of the linearization around (0,0), X'=A1 X, i.e. x'=2x, y'= -y which has a saddle.




```
> dfieldplot([diff(x(t),t)=2*x(t),diff(y(t),t)= -y(t) ],[x(t),y(t)
  ], t=0..1, x=-5..5,y=-5..5);
```




```
> dfieldplot([diff(x(t),t)=2*x(t)-x(t)^2 - x(t)*y(t),diff(y(t),t)=
-y(t) + x(t)*y(t)],[x(t),y(t)], t=0..1, x=1.5..2.5,y=-0.5..0.5);
```

This is the direction field in the box [1.5,2.5]x[-0.5,0.5]. The only equilibrium point in this box is (2,0). It seems that the orbits in this small box looks like the orbits of a linear system with a saddle. In the next figure we will see the direction field of the linearization around (2,0), X'=A2 X, i.e. x'= -2x-2y, y'= y which has a saddle.




```
> dfieldplot([diff(x(t),t)=-2*x(t) - 2*y(t),diff(y(t),t)= y(t) ],[x
  (t),y(t)], t=0..1, x=-5..5,y=-5..5);
```




```
> dfieldplot([diff(x(t),t)=2*x(t)-x(t)^2 - x(t)*y(t),diff(y(t),t)=
-y(t) + x(t)*y(t)],[x(t),y(t)], t=0..1, x=0.5..1.5,y=0.5..1.5);
```

This is the direction field in the box [0.5,1.5]x[0.5,1.5]. The only equilibrium point in this box is (1,1). It seems that the orbits in this small box looks like the orbits of a linear system with an attracting focus. In the next figure we will see the direction field of the linearization around (1,1), X'=A3 X, i.e. x'= -x-y, y'= x which has an attracting focus.




```
> dfieldplot([diff(x(t),t)=-x(t) - y(t),diff(y(t),t)= x(t) ],[x(t),
  y(t)], t=0..1, x=-5..5,y=-5..5);
```




```
> DEplot([diff(x(t),t)=2*x(t)-x(t)^2 - x(t)*y(t),diff(y(t),t)= -y
  (t) + x(t)*y(t)],[x(t),y(t)], t=0..1,[ [x(0)=0.1,y(0)=0.1],[x(0)=
  -0.1,y(0)=0.1],[x(0)=0.2,y(0)=-0.1],[x(0)=-0.1,y(0)=-0.1]]);
```




```
> DEplot([diff(x(t),t)=2*x(t)-x(t)^2 - x(t)*y(t),diff(y(t),t)= -y
  (t) + x(t)*y(t)],[x(t),y(t)], t=0..2,[ [x(0)=1.4,y(0)=0.1],[x(0)=
  2.5,y(0)=0.1],[x(0)=2.5,y(0)=-0.1],[x(0)=1.5,y(0)=-0.1]]);
```




```
> DEplot([diff(x(t),t)=2*x(t)-x(t)^2 - x(t)*y(t),diff(y(t),t)= -y
  (t) + x(t)*y(t)],[x(t),y(t)], t=0..20,[ [x(0)=0.3,y(0)=1.2],[x(0)
  =0.2,y(0)=1.2]]);
```




```
> DEplot([diff(x(t),t)=2*x(t)-x(t)^2 - x(t)*y(t),diff(y(t),t)= -y
  (t) + x(t)*y(t)],[x(t),y(t)], t=0..20,[[x(0)=0.1,y(0)=0.1], [x(0)
  =2.5,y(0)=0.1]]);
```



> In your notebooks, draw the phase portrait of this nonlinear system.

Problem **II.3.** Study the pendulum system x'=y, y'= - 4sin x . This is an idealized model of the motion of the pendulum, since it is assumed that there is no friction with the air.

```
> with(linalg):with(DEtools):with(VectorCalculus):
> subs([x=0,y=0],[y, -4*sin(x)]); eval([y, -4*sin(x)],[x=0,y=0]);
```

Here we check that (0,0) is an equilibrium point.

$$[0, -4\sin(0)]$$

$$[0, 0] \quad \textbf{(9)}$$

```
> Jm3:=Jacobian([ y, -4*sin(x) ],[x,y]);
```

(10)

$$Jm3 := \begin{bmatrix} 0 & 1 \\ -4\cos(x) & 0 \end{bmatrix} \quad (10)$$

```
> A:=subs([x=0,y=0], Jm3);
```
This is the matrix of the linearized system around the equilibrium point (0,0).

$$A := \begin{bmatrix} 0 & 1 \\ -4 & 0 \end{bmatrix} \quad (11)$$

```
> eigenvalues(A);
```
Note that the eigenvalues of A have the real part equal to 0. Thus, the equlibrium point (0,0) is not hyperbolic and, consequently, we can not apply the linearization method. Our next aim is to find a first integral. For this we will write the differential equation of the orbits.

$$2\,\mathrm{I},\ -2\,\mathrm{I} \quad (12)$$

```
> dsolve(diff(y(x),x)=-4*sin(x)/y(x));
```
The differential equation of the orbits is dy/dx=-4sin(x)/y. Note that this is a separable equation, thus we are able to integrate it without Maple, too. Note also that its general solution can be written as y^2-8cos(x)=c, where c is an arbitrary real constant. Thus, H(x,y)=y^2-8cos(x) is a candidate for a global first integral. To assute this, we just have to check the validity in R^2 of the relation y dH/dx -4sin(x) dH/dy=0 (here we have partial derivatives, but I couldnt write the corresponding symbol in Maple).

$$y(x) = \sqrt{8\cos(x) + \_C1},\ y(x) = -\sqrt{8\cos(x) + \_C1} \quad (13)$$

```
> H:=y^2-8*cos(x); y*diff(H,x)-4*sin(x)*diff(H,y);
```

$$H := y^2 - 8\cos(x)$$

$$0 \quad (14)$$

```
> with(plots);
```

[*animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, densityplot, display, dualaxisplot, fieldplot, fieldplot3d, gradplot, gradplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, intersectplot, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra_supported, polyhedraplot, rootlocus, semilogplot, setcolors, setoptions, setoptions3d, shadebetween, spacecurve, sparsematrixplot, surfdata, textplot, textplot3d, tubeplot*] **(15)**

```
> contourplot(y^2-8*cos(x),x=-5..5,y=-5..5);
```
Here we represent the level curves of H. We know that the orbits of the given system lie on the level curves of H. Indeed, we notice closed orbits, which correspond to periodic in time motions of the pendulum. Note also that this is an idealized model of the motion of the pendulum, since it is assumed that there is no friction with the air. This is why periodic motions appear, which can be seen as non-stop oscillations with constant amplitude.


4
y
2
−4
−2
0
2
4
x
−2
−4